

Reporting The Facts: The ODSmemo macro suite for making reproducible RTF memos within SAS®

A. Rocio Lopez, Cleveland Clinic, Cleveland, OH

ABSTRACT

A lot of time and effort is spent on organizing tables and graphics into an easily read and informative report that can be sent to clients. The capability to send formatted text as well as high quality tables and graphics from SAS® to Microsoft Word via the ODS rich text format (RTF) destination allows for easy creation of reproducible statistical reports. This paper will present the ODSmemo macro suite, a set of SAS® macros that permits the user to quickly generate a complete report within the analyst's SAS® code.

INTRODUCTION

Good presentation of statistical analysis is crucial for conveying results to clients. Manually inputting results into a report and editing tables takes time and more importantly increases the potential for errors. The ability to send all these elements directly from SAS® to a word processor allows the user to directly create these reports within SAS®.

This paper will show how the output delivery system (ODS) rich text format can be used to generate reproducible reports and will present a set of macros that can be used to facilitate this process.

ODS RTF

The output delivery system (ODS) rich text format (RTF) destination allows transfer of formatted text and graphics between SAS® and word processing packages such as Microsoft Word.

To open and close the RTF destination the following commands are used:

```
ods rtf file='file_path/file_name.rtf';
    SAS commands here
ods rtf close;
```

In addition to SAS® output, raw text can also be inserted into the RTF file by adding a simple option to the ODS RTF command.

```
ods rtf text="text-string";
```

Formatting and styling of the text and output can be controlled using RTF control words or SAS® in-line formatting.

RTF CONTROL WORDS

A RTF control word is a command that RTF uses to pass information that applications will use to manage documents. A full list of control words can be found in the RTF specification (Microsoft Corporation, 1999).

SAS® allows the use of RTF control words by using the following syntax:

```
^R/RTF' \control_word1\control_word2\...'
```

The following is a list of some control words that can be used in SAS®.

Desired Function	RTF Control Word	Desired Function	RTF Control Word
Italicize	\i	Font size in half points	\fs24
Underline	\ul	Bold	\b
New line	\line	Left aligned	\ql
Bullet	\bullet	Right aligned	\qr
Subscript	\sub	Centered	\qc
Superscript	\super	Tab	\tab
Turn off sub/superscript	\nosupersub	Page break before section	\pagebb

Table 1: RTF Control Words

To turn off features such as italics or bold font add a 0 after the control word.

```
^R/RTF' \b 'This text will be bolded.^R/RTF' \b0 'This text will not.
```

Special characters such as ± or ≥ can be defined using decimal Unicode value; lists of these codes as well as code converters can be found online. Once the code for the desired symbol is known, the following syntax can be used:

```
^R/RTF' \uN\bin'
```

Where **N** is the decimal Unicode value.

INLINE FORMATTING

SAS® inline formatting can also be used to control the style of the RTF file.

```
^S={style-attributes}
```

Where ^ is an escape character that identifies the inline formatting symbol and can be set as follows:

```
ods escapechar="^";
```

Gebhart (2009) provides a good explanation of using inline formatting with ODS markup.

OTHER USEFUL COMMANDS

ODS RTF has many useful options that give more control over the content of the file. For example, the programmer can regulate when to start new pages:

```
ods rtf startpage= yes|no|now;
```

In addition, the use of defined titles and footnotes in ODS GRAPHICS can be easily suppressed;

```
ods rtf nogtitle nogfootnote;
```

The programmer should also keep in mind that while the RTF file remains open all SAS® output will be directed to this destination but the user can control which output is sent to the RTF file with some simple statements:

```
ods rtf select all|none|selections;
```

or

```
ods rtf exclude all|none|selections;
```



A list of all options can be found in the SAS® 9.4 Output Delivery System User's Guide (SAS Institute, 2013).

THE ODSMEMO MACROS SUITE

The ODSmemo macro suite is a set of 5 macros that combines ODS RTF commands to easily produce a statistical report directly from SAS® 9.4, 9.3 or 9.2. It allows the user control over the format of the document as well as inserting free text, tables and figures with the use of some simple commands.

CONSTRUCT HEADER

The %STARTMEMO macro opens the ODS RTF destination and allows the user to set up the header for the report.

```
%StartMemo(  
  HeaderLogo = cc_logo.jpg,   
  HeaderInfo = qhs_header.txt,   
  To = "John Smith, MD|Jane Doe, MD",  
  Department = "Department of Gastroenterology",  
  PreparedBy = "A. Rocio Lopez, MS",  
  Re = "Sample Memo Writing with ODS RTF",  
  Introduction = "This is a sample memo generated using ODSmemo Macro Suite.",  
  RTFFile = /home/lopezr/SAS/MWSUG 2014 Presentation /SampleMemo);
```

Text file with header information. File should have 1 line per line of header and optional email with hyper link can be included

File with logo

The following figure shows the memo header created by these commands.

File destination




Figure 1. Report Header

INCLUDE CONTACT INFORMATION AS FOOTNOTE

The user can also include their personal contact information in the footer area of the report by adding 3 options.

```
%StartMemo (
  HeaderLogo = cc_logo.jpg,
  HeaderInfo = qhs_header.txt,
  To = "John Smith, MD|Jane Doe, MD",
  Department = "Department of Gastroenterology",
  PreparedBy = "A. Rocio Lopez, MS",
  Re = "Sample Memo Writing with ODS RTF",
  Introduction = "This is a sample memo generated using ODSmemo Macro Suite.",
  PersonalFooter = 1,
  Email = lopezr@ccf.org,
  Tel = 012-345-6789,
  RTFFile = /home/lopezr/SAS/MWSUG 2014 Presentation/SampleMemo);
```

The following will be added to your report.




Figure 2. Report Footnote

MANAGE FONTS

%STARTMEMO also allows control over the fonts used through the report as well as line spacing.

```
%StartMemo (
  HeaderLogo = cc_logo.jpg,
  HeaderInfo = qhs_header.txt,
  To = "John Smith, MD|Jane Doe, MD",
  Department = "Department of Gastroenterology",
  PreparedBy = "A. Rocio Lopez, MS",
  Re = "Sample Memo Writing with ODS RTF",
  Introduction = "This is a sample memo generated using ODSmemo Macro Suite.",
  PersonalFooter = 1,
  Email = lopezr@ccf.org,
  Tel = 012-345-6789,
  Font = Times Roman,
  GraphFont = Times Roman,
  TextFontSize = 12,
  TableFontSize = 11,
  GraphFontSize = 11,
  LineSpacing = 1,
  RTFFile = /home/lopezr/SAS/MWSUG 2014 Presentation/SampleMemo);
```

Must be an acceptable SAS font

ADD PARAGRAPHS

Following the initial report setup, the %ADDSECTION macro allows the user to add sections of text to the report. A call to the macro will look as follows:

```
1 %AddSection(
  Header= "Aim.", → Optional section header
  Text= "The aim of this memo is to show the use of the ODSmemo Macro Suite.");
```

USE OF SPECIAL CHARACTERS

RTF control words can be used to insert special characters into the report. Commonly used symbols and functions have been defined within the macros as macro parameters and can be called by the user at any time.

```
2 %AddSection(
  Header="Statistical Analysis.",
  Text="Descriptive data are presented as means &PIMi. SD or N (%).
  A linear regression analysis was performed to study the association
  between height and weight. A p < 0.05 was considered statistically
  significant. All analyses were performed using SAS version 9.3 (The SAS
  Institute, Cary, NC) and R version 3.0.2 (The R Foundation for
  Statistical Computing, Vienna, Austria) was used to generate Figure1.");
```

The following is a list of the pre-defined macro parameters that can be called by the user.

Macro Parameter	Purpose
NewLine	Insert new line
Tab	Insert tab
Bold	Turn on bold font
BoldOff	Turn off bold font
Underline	Turn on underline
UnderlineOff	Turn off underline
Italics	Turn on italics
ItalicsOff	Turn off italics
Superscript	Superscript following text.
Subscript	Subscript following text

Macro Parameter	Purpose
Super_Sub_Off	Turn off superscript/subscript
Bullet	Insert bullet point
TabBullet	Insert indented bullet point
PabeBB	Page break before section
Le	Insert ≤ symbol
Ge	Insert ≥ symbol
NotEqual	Insert ≠ symbol
Mu	Insert μ symbol
Alpha	Insert α symbol
PIMi	Insert ± symbol

Table 2: Macro parameters defined for functions and symbols

INCORPORATING SAS® RESULTS WITH USE OF MACRO PARAMETERS

The user can also create macro parameters containing key results from their analysis to be automatically inserted in the text; this allows for easy reproducibility of the report.

```

3 %AddSection (
  Header="Results."
  Text="A total of &N. subjects were included in the analysis, Table 1 presents
  a summary of patient characteristics. Height was found to be significantly
  associated with weight (Table 2 and Figure 1). For every 1 inch increase in
  height, the average weight increases by approximately &HtBeta. lbs (p &p.).");
  
```

User defined macro parameter

Below is a look at the sections added to the report up to this point.




Figure 3. Adding Paragraphs to Report

ADD TABLES

User generated macros can be used to insert tables into the report but beware of formatting options of each macro and how these can impact the look of the tables in the ODS RTF destination.

Additionally, the %ADDTABLE macro allows the user to insert a formatted table generated from a SAS® data set. The data should be properly labeled and formatted.

```

%AddTable (
  ds= Table2,
  Title= "Table 2: Linear Regression Analysis",
  LabelCol= Variable,
  Columns= Estimate SE pfmt,
  Footnote= "Adjusted R2 &superscript.2&super_sub_off. = &AdRSq.&newline.
  SE:standard error",
  Cwidth1= 150,
  Cwidth2= 150);
  
```

Width of LabelCol

Width of Columns




Figure 4. Adding Tables to Report

ADD FIGURES

The %ADDFIGURE macros will allow the user to insert external figures into the report; allowable file extensions are .png or .jpeg.

```
%AddFigure (
  Title= "&boldtext1.Figure 1: &boldoff.External figure imported with AddFigure
  macro",
  FigFile= /home/lopezr/SAS/MWSUG_2014_Presentation/SampleMemo/Figure1.png,
  NewPage= 1);
```

Annotations: 'Title' points to the Title parameter, 'Start new page before figure' points to NewPage= 1, and 'File path' points to the FigFile parameter.

Alternatively, SAS® graphics can be easily included using ODS graphics or SAS® graphics procedures such as The SGPLOT procedure.

%CLOSEMEMO

The %CLOSEMEMO macro will close the ODS RTF destination and populate your file. Only after submitting this command will the user be able to access their final report.

EXAMPLE

The following is an example of a report generated using the ODSmemo macros suite; code used to produce this memo is found in Appendix 1.

Date: August 25, 2014

To: John Smith, MD
Jane Doe, MD
Department of Gastroenterology

From: A. Rocio Lopez, MS

Re: Sample Memo Writing with ODS RTF

This is a sample memo generated using ODSmemo Macro Suite.

Aim.

The aim of this memo is to show the use of the ODSmemo Macro Suite.

Statistical Analysis.

Descriptive data are presented as means \pm SD or N (%). A linear regression analysis was performed to study the association between height and weight. A $p < 0.05$ was considered statistically significant. All analyses were performed using SAS version 9.3 (The SAS Institute, Cary, NC) and R version 3.0.2 (The R Foundation for Statistical Computing, Vienna, Austria) was used to generate Figure 1.

Results.

A total of 19 subjects were included in the analysis, Table 1 presents a summary of patient characteristics. Height was found to be significantly associated with weight (Table 2 and Figure 1). For every 1 inch increase in height, the average weight increases by approximately 3.9 lbs ($p < 0.001$).

Figure 5a. Sample output from ODSmemo macro suite: Page 1

Table 1 - Patient Characteristics

Factor	Total (N=19)
Sex	
. F	9(47.4)
. M	10(52.6)
Age	13.3±1.5
Height	62.3±5.1
Weight	100.0±22.8

Values presented as Mean ± SD, Median [P25, P75],
Median (min, max) or N (column %).

Table 2: Linear Regression Analysis

Factor	Estimate (SE)	p-value
Intercept	-143.0 (32.3)	<0.001
Height	3.9 (0.52)	<0.001

Adjusted R² = 0.76.
SE: standard error.

Figure 5b. Sample output from ODSmemo macro suite: Page 2

Figure 1: External figure imported with AddFigure macro




Figure 2: Figure Generated Using PROC SGPLOT




Figure 5c. Sample output from ODSmemo macro suite: Page 3

CONCLUSION

ODS is a useful and powerful tool for directing SAS® output using a reader friendly format. The ODSmemo macros suite takes advantage of these features and is a way to easily generate reproducible statistical reports.

FULL SOURCE CODE

You can download the ODSmemo macro suite at <http://www.sascommunity.org/wiki/User:Lopezr>.

REFERENCES

- Microsoft Corporation (1999), Rich Text Format (RTF) Specification, version 1.6. [http://msdn.microsoft.com/en-us/library/aa140277\(v=office.10\)](http://msdn.microsoft.com/en-us/library/aa140277(v=office.10)).
- Gebhart, Eric. 2009. "InlineFormatting with ODS Markup." *Proceedings of the SAS Global Forum 2009 Conference*. Available at <http://support.sas.com/resources/papers/proceedings09/222-2009.pdf>.
- SAS Institute (2013), SAS® 9.4 Output Delivery System: User's Guide. Second Edition. <http://support.sas.com/documentation/cdl/en/odsug/66611/HTML/default/viewer.htm#p1vsv8ucnjzinn1wq5wrlp74mdb.htm>.

ACKNOWLEDGMENTS

Thanks to Susana Arrigain and Matt Karafa for reviewing the paper.

RECOMMENDED READING

- SAS®9 ODS RTF Tip Sheet: <http://support.sas.com/rnd/base/ods/odsrtf/rtf-tips.pdf>.
- Haworth Lauren. 2011. "ODS RTF: the Basics and Beyond". *Proceedings of the SAS Global Forum 2011 Conference*. Available at <http://support.sas.com/resources/papers/proceedings11/263-2011.pdf>.
- Parsons, Lori. 2007. "Enhancing RTF Output with RTF Control Words and In-Line Formatting". *Proceedings of the SAS Global Forum 2007 Conference*. Available at <http://www2.sas.com/proceedings/forum2007/151-2007.pdf>.
- Rachabattula, Sriharsha. 2010. "Using RTF codes in ODS RTF outputs." *Proceedings of the Northeast SAS User Group (NESUG) 2010 Conference*. Available at <http://www.nesug.org/Proceedings/nesug10/po/po40.pdf>.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: A. Rocio Lopez
Enterprise: Cleveland Clinic
Address: 9500 Euclid Ave/JJN3-01
City, State ZIP: Cleveland, OH, 44195
E-mail: lopezr@ccf.org

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

APPENDIX 1. CODE USED TO GENERATE EXAMPLE REPORT

```
/******  
  
                *Basic Setup;  
  
/******  
*****;  
*options  
*****;  
options pagesize=60 ls=78 nodate nonumber nocenter formdlm='~'  
        sasautos = ('/home/rlopez/SASmacros')  
        /*sasautos option specifies location of autocall library*/  
        implmac mautosource mrecall;  
  
*****;  
*include formats  
*****;  
proc format;  
    picture Pvaluef (round)  
        0.985 - high = "0.99" (NoEdit)  
        0.10  -< 0.985 = "9.99"  
        0.001 -< 0.10 = "9.999"  
        0     -< 0.001 = "<0.001" (NoEdit)  
        . = " ";  
run;  
quit;  
  
*****;  
*Set working directory  
*****;  
%let BaseDir=/home/lopezr/SAS/MWSUG 2014 Presentation;  
  
%include "&BaseDir./ODSMemoMacroSuite/startmemo.sas";  
  
/******  
  
                *Using sashelp.class --DB Contents;  
  
/******  
proc contents data=sashelp.class varnum;  
    run;  
  
/******  
  
                *Analysis;  
  
/******  
  
*****;  
**Generate Table 1-Patient Characteristics;  
*****;  
%summtable(data=sashelp.class,  
cat2 = Sex,  
con1 = Age Height Weight,  
sortby = _list,  
list = Sex Age Height Weight,  
tbltitle = Table 1 - Patient Characteristics,  
rtfout = Table1);  
  
/**Note: This macro can also be downloaded from  
http://www.sascommunity.org/wiki/User:Lopezr and has an option that  
allows it to be called within the ODSMemo macro suite (ODSMemo=1)**/  
*****
```

```

*****;
**Generate Table 2 - Linear Regression Analysis;
*****;
**Run linear regression analysis;
proc reg data=sashelp.class;
    model weight = height;
    ods output ParameterEstimates=PEs FitStatistics=Fit;
run;

**Create macro variable with Adjusted R-Square value;
proc sql noprint;
    select round(nvalue2,0.01) into :AdjRSquare
    from Fit where Label2="Adj R-Sq";

**Save all values that will be included in the memo;
data Table2;
    retain Variable Estimate_SE ProbT;
    set PEs;

    informat Estimate_SE $25.;

    Estimate_SE=compress(put(Estimate, __estf.)||" ("||
        compress(put(StdErr, __estf.)||")");

    if _n_=1 then AdjRSquare=&AdjRSquare;

    pfmt=probt;

    format pfmt pvaluef.;

    label Variable="Factor"
        Estimate_SE="Estimate (SE)"
        pfmt="p-value";

    keep Variable Estimate_SE ProbT Estimate StdErr AdjRSquare pfmt;

run;

proc print data= Table2 noobs; run;

*****;
**Create macro variables for values will be called in the macros;
*****;
data _null_;
    set Table1;
    if _n_=1;
    call symput('N', trim(left(put(_N1, 9.0)))); **number of subjects;

data _null_;
    set Table2;

    if Variable="Intercept" then do;
        call symput('Intercept', trim(left(put(Estimate, 8.1)))); **intercept;
        call symput('AdRSq', trim(left(put(AdjRSquare, 8.2)))); **adjusted r^2;
    end;

    if Variable="Height" then do;
        call symput('HtBeta', trim(left(put(Estimate, 8.1)))); **slope for height;
        call symput('p', trim(left(put(ProbT, pvaluef.)))); **p-value for height;
    end;

run;

```

```

/*****/

                *Generate Report using ODSMemo macro suite ;

/*****/
*****;
**Memo set-up          ;
*****;
%StartMemo(
HeaderLogo = cc_logo.jpg,
HeaderInfo = qhs_header.txt,
To = "John Smith, MD|Jane Doe, MD",
Department = "Department of Gastroenterology",
PreparedBy = "A. Rocio Lopez, MS",
Re = "Sample Memo Writing with ODS RTF",
Introduction = "This is a sample memo generated using ODSmemo Macro Suite.",
PersonalInfoFooter = 1,
Email = lopezr@ccf.org,
Tel = 012-315-6789,
Font = Times Roman,
GraphFont = Times Roman,
TextFontSize = 12,
TableFontSize = 11,
GraphFontSize = 11,
LineSpacing = 1 ,
RTFFile = &BaseDir./SampleMemo);

*****;
**Aims Section          ;
*****;
%AddSection(
Header="Aim.",
Text= "The aim of this memo is to show the use of the ODSmemo Macro Suite.");

*****;
**Statistical Analysis Section;
*****;
%AddSection(
Header="Statistical Analysis.",
Text="Descriptive data are presented as means &PLMi. SD or N (%).
A linear regression analysis was performed to study the association
between height and weight. A p < 0.05 was considered statistically
significant. All analyses were performed using SAS version 9.3 (The SAS
Institute, Cary, NC) and R version 3.0.2 (The R Foundation for
Statistical Computing, Vienna, Austria) was used to generate Figure1.");

*****;
**Results Section      ;
*****;
%AddSection(
Header="Results.",
Text="A total of &N. subjects were included in the analysis, Table 1 presents
a summary of patient characteristics. Height was found to be significantly
associated with weight (Table 2 and Figure 1). For every 1 inch increase in
height, the average weight increases by approximately &HtBeta. lbs (p &p.).");

*****;
**Add table using other macro ;
*****;
**Start new page;
ods rtf startpage=now;

```

```

**Call macro;
%summtable(data=sashelp.class,
cat2 = Sex,
con1 = Age Height Weight,
sortby = _list,
list = Sex Age Height Weight,
prints = 0,
tbltitle = Table 1 - Patient Characteristics,
ODSMemo = 1);

*****;
**Add table using AddTable ;
*****;
**Insert more blank space between tables;
ods text="^R/RTF'\line\line '";

%AddTable(
ds = Table2,
Title = "Table 2: Linear Regression Analysis",
LabelCol = Variable,
Columns = Estimate_SE pfmt,
Footnote = "Adjusted R2&super_sub_off. = &AdRSq.&newline.SE:
standard error.",
cwidth1 = 150,
cwidth2 = 150);

*****;
**Add figure using AddFigure ;
*****;
%AddFigure(
Title = "&boldtext1.Figure 1: &boldoff.External figure imported with
AddFigure macro",
NewPage = 1,
FigFile = &BaseDir./Figure1.png);

*****;
**Add figure using SGPLOT ;
*****;
/*Note: This will always be center justified. To left justify, you would need
to create the png file and call it back in with GREPLAY or Figure macros*/

**Add Figure Title;
ods rtf Text = "^S={outputwidth=100% just=1}&newline2.&boldtext1.
Figure 2:&boldoff. Figure Generated Using PROC SGPLOT";

**Change resolution, do not include footnotes on plot;
ods rtf image_dpi=500 nogfootnote;

**Set size of figure;
ods graphics / noborder width=3in height=3in;

**Generate figure;
proc sgplot data=sashelp.class noautolegend;

    **Scatter plot of height and weight;
    scatter x=height y=weight
        /markerattrs=(color=black symbol="CircleFilled");

    **Add regression line;
    lineparm x=0 y=&Intercept. slope=&HtBeta
        /lineattrs=(color=black);

    **Add text with regression model, p and R^s;

```

```
inset "Weight = &Intercept. + Height * &HtBeta"
      "p-value &p."
      "Adjusted R(*ESC*){sup '2'} = &AdRSq"
      /textattrs=(family=Times size=9);

run;

*****;
**Close and generate memo      ;
*****;
%CloseMemo;
```